Maple Commands for Surfaces and Partial Derivatives

We've seen that a curve can be written as $y = f(x)$, or more generally in parametric form using one parameter (usually t), and the curve can be in two or three dimensions:

$$\mathbf{r}(t) = \langle x(t), y(t) \rangle \qquad \mathbf{r}(t) = \langle x(t), y(t), z(t) \rangle$$

Similarly, a surface can be represented in simple terms as the graph of $z = f(x, y)$, or in parametric form using *two* parameters, for example, u, v:

$$\mathbf{F}(u, v) = \langle x(u, v), y(u, v), z(u, v) \rangle$$

Maple can plot either using the `plot3d` command.

Examples

- Plot the upper hemisphere of a sphere of radius 1.
  - As a surface $x^2+y^2+z^2 = 1$, we can solve for z and take the positive root. The first option, `axes=NORMAL` puts the axes on the graph (a second option to try might be `axes=BOXED`). Sometimes Maple scales the graph- To make the scaling look right (a circle as a circle and not an ellipse), we use the option `scaling=CONSTRAINED`. The line is broken to fit in the margin- Don't break the line in Maple.

    ```
    plot3d(sqrt(1-x^2-y^2),x=-1.5..1.5,y=-1.5..1.5,
                           axes=NORMAL, scaling=CONSTRAINED);
    ```

  - As a parametrized surface, any point on the sphere can be determined by two angles: θ, ϕ (ϕ is read as "fee"). In this case, ϕ is the angle formed from the "north pole" to the point (x, y, z), and θ is the angle the point $(x, y, 0)$ makes in the xy plane (in the usual sense).

    $$x = \cos(\theta)\sin(\phi) \qquad y = \sin(\theta)\sin(\phi) \qquad z = \cos(\phi)$$

    (We'll learn how to construct these next time)

    ```
    plot3d([cos(theta)*sin(phi),sin(theta)*sin(phi),cos(phi)],
                            theta=0..2*Pi, phi=0..Pi/2,scaling=CONSTRAINED);
    ```

    For fun, you might try changing these parameters to see if you can determine what they control in the graph.

- Plot the surface $z = x\mathrm{e}^{-x^2-y^2}$, and color the graph according to $\sin(xy)$:

  ```
  plot3d(x*exp(-x^2-y^2), x = -2..2, y =-2..2, color = sin(x*y))
  ```

  Just for fun, we can also do an animation, where we vary the height of the valley and hilltop by using an extra parameter A:

```
with(plots):
animate( plot3d, [sin(A)*x*exp(-x^2-y^2), x=-2..2, y=-2..2],
                   A=0..2*Pi,shading=zhue );
```

Then click on the graph and there will be a play button on the menu. There is also an option to run a loop- See if you can find it.

- Plot two surfaces:

  Method 1: Plot the two separately, then "display" the results. Use this technique if one graph is not parametric, and one is- Here is an example.

```
with(plots):
A:=plot3d(sin(x)*sin(y),x=-Pi..Pi,y=-Pi..Pi,color=white):
B:=plot3d([cos(theta)*sin(phi),sin(theta)*sin(phi),cos(phi)],
             theta=0..2*Pi, phi=0..Pi/2,scaling=CONSTRAINED):
display({A,B});
```

  Method 2: If your functions are all of the form $z = f(x, y)$, you can use a single plot command. For example, here we plot $\sin(xy)$, $x - y$ and $4e^{-x^2-y^2}$ all on the same graph:

```
plot3d({sin(x*y), x-y, 4*exp(-x^2-y^2)},x=-2..2,y=-2..2);
```

Multivariate Limits

Multivariate limits can be difficult to compute by hand- Maple can have a difficult time as well. However, often Maple is able to compute a multivariate limit- and we use the `limit` command we learned earlier. There are some excellent examples in the help file: `?limit`

Derivatives and Partial Derivatives

Partial derivatives are easy in Maple, as are higher derivatives.

EXAMPLE: Given $f(x, y) = \sqrt{x^2 + y^2}$, find f_x, f_{yx} and f_{yyx}

SOLUTION: (I try to make my variable names make sense in this context)

```
f:=sqrt(x^2+y^2);
fx:=diff(f,x);
fyx:=diff(f,y,x);
fyyx:=diff(f,y,y,x);
```

Is f_{yyx} the same as f_{yxy}? Check in Maple.